

Projekt: Werkstückanschlag

Aufgabe: Nadelhalter Pos. 4


Programmierung: Heidenhain iTNC (Klartext)

Autor:

Klasse:

Datum:

```
0 BEGIN PGM Nadelh_4 MM
1 BLK FORM 0.1 Z X-20 Y-20 Z-22
2 BLK FORM 0.2 X+20 Y+20 Z+0
3 ;
4 ;Werkzeug SFR VHM D=20mm
5 ;Werkzeugaufruf
6 TOOL CALL 10 Z S8000 F1000
7 M3
8 ;
9 L X-30 Y-30 R0 FMAX
10 L Z+2 R0 FMAX
11 L Z-15
12 ;
13 APPR LT X-15 Y-15 LEN10 RL
14 L Y+15
15 L X+15
16 L Y-15
17 L X-15
18 DEP LT LEN10
19 L X-30 Y-30 R0
20 L Z+2 R0 FMAX
21 ;
22 L X+0 Y+0 R0 FMAX
23 L Z+2 R0 FMAX
24 ;
25 CYCL DEF 257 KREISZAPFEN ~
    Q223=+14.959 ;FERTIGTEIL-DURCHM. ~
    Q222=+22 ;ROHTEIL-DURCHMESSER ~
    Q368=+0.2 ;AUFMASS SEITE ~
    Q207=+500 ;VORSCHUB FRAESEN ~
    Q351=+1 ;FRAESART ~
    Q201=-3 ;TIEFE ~
    Q202=+5 ;ZUSTELL-TIEFE ~
    Q206=+1000 ;VORSCHUB TIEFENZ. ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q203=+0 ;KOOR. OBERFLAECHE ~
    Q204=+50 ;2. SICHERHEITS-ABST. ~
    Q370=+1 ;BAHN-UEBERLAPPUNG
26 ;
27 CYCL CALL POS X+0 Y+0 Z+0
28 ;
29 L Z+2 R0 FMAX
30 L X-20 Y-20 R0 FMAX
```



```
31 L Z-3 R0 F AUTO
32 APPR LT X-5 Y-10 LEN5 RL
33 L Y+10
34 DEP LT LEN5
35 L X-20 Y-20 R0 F AUTO
36 L Z+50 R0 FMAX
37 ;
38 ;Werkzeug NCANBO D=12mm
39 ;Werkzeugaufruf
40 TOOL CALL 39 Z S6000 F500
41 M3
42 ;
43 L X+0 Y+0 R0 FMAX
44 L Z+2 R0 FMAX
45 ;
46 CYCL DEF 240 ZENTRIEREN ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q343=+1 ;AUSWAHL DURCHM/TIEFE ~
    Q201=-2 ;TIEFE ~
    Q344=-9.3 ;DURCHMESSER ~
    Q206=+150 ;VORSCHUB TIEFENZ. ~
    Q211=+0.5 ;VERWEILZEIT UNTEN ~
    Q203=+0 ;KOOR. OBERFLAECHE ~
    Q204=+50 ;2. SICHERHEITS-ABST.
47 ;
48 CYCL CALL POS X+0 Y+0 Z+0
49 ;
50 ;Werkzeug SPIBO D=8.5mm
51 ;Werkzeugaufruf
52 TOOL CALL 34 Z S6000 F500
53 M3
54 ;
55 CYCL DEF 205 UNIVERSAL-TIEFBOHREN ~
    Q200=+2 ;SICHERHEITS-ABST. ~
    Q201=-17 ;TIEFE ~
    Q206=+150 ;VORSCHUB TIEFENZ. ~
    Q202=+5 ;ZUSTELL-TIEFE ~
    Q203=+0 ;KOOR. OBERFLAECHE ~
    Q204=+50 ;2. SICHERHEITS-ABST. ~
    Q212=+0 ;ABNAHMEBETRAG ~
    Q205=+0 ;MIN. ZUSTELL-TIEFE ~
    Q258=+0.2 ;VORHALTEABSTAND OBEN ~
    Q259=+0.2 ;VORHALTEABST. UNTEN ~
    Q257=+0 ;BOHRTIEFE SPANBRUCH ~
    Q256=+0.2 ;RZ BEI SPANBRUCH ~
    Q211=+0 ;VERWEILZEIT UNTEN ~
    Q379=+0 ;STARTPUNKT ~
    Q253=+750 ;VORSCHUB VORPOS.
56 ;
57 CYCL CALL POS X+0 Y+0 Z+0
58 ;
59 ;Werkzeug SFR VHM D=4mm
60 ;Werkzeugaufruf
61 TOOL CALL 2 Z S10000 F1000
62 M3
63 ;
64 L X-10 Y-21 R0 FMAX
65 L Z+2 R0 FMAX
66 L Z-6 R0 F AUTO
67 ;
68 L Y+21
69 L Z+50 R0 FMAX
```

```
70 ;  
71 ;Werkzeug Fasenfr D=6mm  
72 ;Werkzeugaufruf  
73 TOOL CALL 41 Z S10000 F800  
74 M3  
75 ;  
76 L X-8.5 Y-21 R0 FMAX  
77 L Z+2 R0 FMAX  
78 L Z-5.5 R0 F AUTO  
79 L Y+21  
80 L X-11.5  
81 L Y-21  
82 L Z+50 R0 FMAX  
83 ;  
84 TOOL CALL 0 Z  
85 ;  
86 M30  
87 END PGM Nadelh_4 MM
```